

PBCC Y2K Training Manual — UNISYS COBOL Program Conversion & Testing Procedures

1. Overview

2. Standard “Y2K” Processing
COBOL Copyprocs

3. Standard “Y2K” Processing
Minimum Requirements

4. “IF” Statements

5. “SUBTRACT” Statements

6. “COMPUTE” Statements

7. Julian Date Processing

8. Existence Checking
Range Checking

9. ONLINE Program Considerations

10. Testing Y2K Compliance

11. Appendix A — Sample Shell Script
— Unit Test Output

12. Appendix B — Additional Notes

Standard “Y2K” Processing — COBOL Copyprocs

Using “COPYPROC” (code insert) within Unisys COBOL:

Name Of PROC	Placement In COBOL Source	Usage In COBOL Source Code	Contents & Notes
WS-DATES	Start of WORKING STORAGE	Required Fields for Y2K	Handles All PBCC Date Formats
98-WINDOW	End of COBOL Source Code	Date Expansion Range Checks (9's) Existence Checks (0's)	Conforms to Pitney Bowes Compliance Standards
99-PIVOT	End of COBOL Source Code	Date Windowing. May be necessary for Database inserts/sorting.	Not Applicable as of July 1999

Beginning on the next page, are listings of COBOL copyprocs “**WS-DATES**” and “**98-WINDOW**”.

Listing of Copyproc WS-DATES

WS-DATES PROC

Y2000 01 WS-DATE-1.

```

Y2000 05 WS-CCYYMMDD-1 PIC 9(08) VALUE ZEROS.
Y2000 05 WS-FILLER-A REDEFINES WS-CCYYMMDD-1.
Y2000 10 WS-CC-1 PIC X(02).

Y2000 10 WS-YY-1 PIC X(02).
Y2000 10 WS-MM-1 PIC X(02).
Y2000 10 WS-DD-1 PIC X(02).
Y2000 05 WS-FILLER-B REDEFINES WS-CCYYMMDD-1.

Y2000 10 WS-CCYYMM-1 PIC 9(06).
Y2000 10 FILLER PIC X(02).
Y2000 05 WS-FILLER-C REDEFINES WS-CCYYMMDD-1.

Y2000 10 WS-CCYY-1 PIC 9(04).
Y2000 10 FILLER PIC X(04).
Y2000 05 WS-FILLER-D REDEFINES WS-CCYYMMDD-1.
Y2000 10 FILLER PIC X(01).

Y2000 10 WS-CYYMDD-1 PIC X(07).
Y2000 05 WS-FILLER-E REDEFINES WS-CYYMDD-1.
Y2000 10 FILLER PIC X(02).

Y2000 10 WS-YMDD-1 PIC X(06).
Y2000 05 WS-FILLER-F REDEFINES WS-CYYMDD-1.
Y2000 10 FILLER PIC X(02).

Y2000 10 WS-YMM-1 PIC X(04).
Y2000 10 FILLER PIC X(02).
Y2000

```

Output [CCYYMMDD]

Input [Example: "000101"]

Y2000 01 WS-DATE-2.

```

Y2000 05 WS-CCYYMMDD-2 PIC 9(08) VALUE ZEROS.
Y2000 05 WS-FILLER-G REDEFINES WS-CCYYMMDD-2.
Y2000 10 WS-CC-2 PIC X(02).

Y2000 10 WS-YY-2 PIC X(02).
Y2000 10 WS-MM-2 PIC X(02).
Y2000 10 WS-DD-2 PIC X(02).
Y2000 05 WS-FILLER-H REDEFINES WS-CCYYMMDD-2.

Y2000 10 WS-CCYYMM-2 PIC 9(06).
Y2000 10 FILLER PIC X(02).
Y2000 05 WS-FILLER-I REDEFINES WS-CCYYMMDD-2.

Y2000 10 WS-CCYY-2 PIC 9(04).
Y2000 10 FILLER PIC X(04).
Y2000 05 WS-FILLER-J REDEFINES WS-CCYYMMDD-1.
Y2000 10 FILLER PIC X(01).

Y2000 10 WS-CYYMDD-2 PIC X(07).
Y2000 05 WS-FILLER-K REDEFINES WS-CYYMDD-2.
Y2000 10 FILLER PIC X(02).

Y2000 10 WS-YMDD-2 PIC X(06).
Y2000 05 WS-FILLER-L REDEFINES WS-CYYMDD-2.
Y2000 10 FILLER PIC X(02).

Y2000 10 WS-YMM-2 PIC X(04).
Y2000 10 FILLER PIC X(02).
Y2000

```

Output [CCYYMM]

Input [Example: "0211"]

Y2000 01 WS-DATE-3.

```

Y2000 05 WS-MMDDYY-3 PIC 9(06) VALUE ZEROS.
Y2000 05 WS-FILLER-M REDEFINES WS-MMDDYY-3.

Y2000 10 WS-MM-3 PIC X(02).

Y2000 10 WS-DD-3 PIC X(02).

Y2000 10 WS-YY-3 PIC X(02).

```

Input

Note: Output either
WS-CCYYMMDD-1
WS-CCYYMMDD-2

(Continued Next Page)

Listing of Copyproc WS-DATES

(Continued)

Y2000					
Y2000 01 WS-DATE-4.					
Y2000	05	WS-CCYYMMCY-4	PIC 9(07)	VALUE ZEROS.	Output
Y2000	05	WS-FILLER-N REDEFINES WS-CCYYMMCY-4.			[CCYYMMCY]
Y2000	10	WS-CC-4	PIC X(02).		
Y2000	10	WS-YY-4	PIC X(02).		
Y2000	10	WS-MM-4	PIC X(02).		
Y2000	10	WS-CY-4	PIC X(01).		
Y2000	05	WS-FILLER-O REDEFINES WS-CCYYMMCY-4.			
Y2000	10	FILLER	PIC X(02).		
Y2000	10	WS-YYMMCY-4	PIC 9(05).		Input
Y2000					[CY = 1-digit Cycle]
Y2000 01 WS-DATE-5.					
Y2000	05	WS-CCYYMMCY-5	PIC 9(07)	VALUE ZEROS.	Output
Y2000	05	WS-FILLER-P REDEFINES WS-CCYYMMCY-5.			[CCYYMMCY]
Y2000	10	WS-CC-5	PIC X(02).		
Y2000	10	WS-YY-5	PIC X(02).		
Y2000	10	WS-MM-5	PIC X(02).		
Y2000	10	WS-CY-5	PIC X(01).		
Y2000	05	WS-FILLER-Q REDEFINES WS-CCYYMMCY-5.			
Y2000	10	FILLER	PIC X(02).		
Y2000	10	WS-YYMMCY-5	PIC 9(05).		Input
Y2000					[Example: "01064"]
Y2000 01 WS-DATE-6.					
Y2000 05 WS-YYMM-6.					
Y2000	10	WS-YY-6-COMP	PIC S9(05)	VALUE ZEROS.	Input
Y2000	10	WS-FILLER-R REDEFINES WS-YY-6-COMP.			
Y2000	15	FILLER	PIC X(03).		
Y2000	15	WS-YY-6	PIC S9(02).		Output
					[Move to WS-YY-1 or WS-YY-2]
Y2000	10	WS-MM-6-COMP	PIC S9(05)	VALUE ZEROS.	Input
Y2000	10	WS-FILLER-S REDEFINES WS-MM-6-COMP.			
Y2000	15	FILLER	PIC X(03).		
Y2000	15	WS-MM-6	PIC S9(02).		Output
Y2000					[Move to WS-MM-1 or WS-MM-2]
Y2000 01 WS-DATE-7.					
Y2000	05	WS-CCYYDDD-7	PIC 9(07)	VALUE ZEROS.	Output
Y2000	05	WS-FILLER-T REDEFINES WS-CCYYDDD-7.			[CCYYDDD]
Y2000	10	WS-CC-7	PIC 9(02).		
Y2000	10	WS-YYDDD-7	PIC 9(05).		Input
Y2000	05	WS-FILLER-U REDEFINES WS-CCYYDDD-7.			[Julian Date]
Y2000	10	FILLER	PIC 9(02).		
Y2000	10	WS-YY-7	PIC 9(02).		
Y2000	10	WS-DDD-7	PIC 9(03).		
Y2000					
Y2000 01 WS-DATE-8.					
Y2000	05	WS-CCYYDDD-8	PIC 9(07)	VALUE ZEROS.	Output
Y2000	05	WS-FILLER-V REDEFINES WS-CCYYDDD-8.			[CCYYDDD]
Y2000	10	WS-CC-8	PIC 9(02).		
Y2000	10	WS-YYDDD-8	PIC 9(05).		Input
Y2000	05	WS-FILLER-W REDEFINES WS-CCYYDDD-8.			[Example: "00235"]
Y2000	10	FILLER	PIC 9(02).		
Y2000	10	WS-YY-8	PIC 9(02).		
Y2000	10	WS-DDD-8	PIC 9(03).		

(Continued Next Page)

Date Existence & Validity Checks

Y2000				
Y2000	77	WS-CENTURY	PIC 9(02)	VALUE ZEROS.
Y2000	88	VALID-CENTURY	VALUES 19 THRU 20.	
Y2000	88	LOW-CENTURY	VALUES 01.	
Y2000	88	HIGH-CENTURY	VALUES 99.	
Y2000	77	WS-YEAR	PIC 9(02).	
Y2000	88	VALID-YEAR	VALUES 00 THRU 99.	
Y2000	88	LOW-YEAR	VALUES 00.	
Y2000	88	HIGH-YEAR	VALUES 99.	
Y2000	77	WS-MONTH	PIC 9(02).	
Y2000	88	VALID-MONTH	VALUES 01 THRU 12.	
Y2000	88	LOW-MONTH	VALUES 00.	
Y2000	88	HIGH-MONTH	VALUES 99.	
Y2000	77	WS-DAY	PIC 9(02).	
Y2000	88	VALID-DAY	VALUES 01 THRU 31.	
Y2000	88	LOW-DAY	VALUES 00.	
Y2000	88	HIGH-DAY	VALUES 99.	
Y2000				
Y2000	77	WS-DAYS	PIC 9(03).	
Y2000	88	VALID-DAYS	VALUES 001 THRU 732.	
Y2000	88	LOW-DAYS	VALUES 000.	
Y2000	88	HIGH-DAYS	VALUES 999.	

(Julian Days)

Y2K Reserved (Database)

Y2000				
Y2000	77	WS-PIVOT	PIC 9(02).	
Y2000	77	WS-TABLE	PIC 9(03).	

Y2K Reserved (**)

Y2000				
Y2000	77	WS-A	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-B	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-C	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-D	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-E	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-F	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-G	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-H	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-I	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-J	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-K	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-L	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-M	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-N	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-O	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-P	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-Q	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-R	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-S	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-T	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-U	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-V	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-W	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-X	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-Y	PIC 9(01)	VALUE ZEROS.
Y2000	77	WS-Z	PIC 9(01)	VALUE ZEROS.
END				

(**) In "local" COBOL Working Storage, previous use of "WS-A" through "WS-Z" will result in **SEV. M** compile errors. As a result, Y2K has adopted the following standards:

1. **Program** = **WS-**<program name>
2. **Character** = **WSC-**<literal>
3. **String** = **WSL-**<literal string>
4. **Numerics** = **WSN-**<numeric value>.

Listing of Copyproc 98-WINDOW

98-WINDOW PROC

```
Y2000 99998-IF-YY-1.
Y2000
Y2000 MOVE ZEROS TO WS-CC-1.
Y2000
Y2000 MOVE WS-YY-1 TO WS-YEAR.
Y2000
Y2000 IF (LOW-CENTURY)
Y2000 AND (LOW-YEAR)
Y2000 MOVE 01 TO WS-CC-1
Y2000 GO TO 99998-EXIT.
Y2000
Y2000 IF (HIGH-CENTURY)
Y2000 AND (HIGH-YEAR)
Y2000 MOVE 99 TO WS-CC-1
Y2000 GO TO 99998-EXIT.
Y2000
Y2000 IF WS-YY-1 > 40
Y2000 MOVE 19 TO WS-CC-1
Y2000 ELSE
Y2000 MOVE 20 TO WS-CC-1.
Y2000
Y2000 IF VALID-YEAR
Y2000 GO TO 99998-EXIT.
Y2000
Y2000 MOVE ZEROS TO WS-CCYY-1.
Y2000 GO TO 99998-EXIT.
Y2000
```

YY to CCYY

```
Y2000 99998-IF-YY-2.
Y2000
Y2000 MOVE ZEROS TO WS-CC-2.
Y2000
Y2000 MOVE WS-YY-2 TO WS-YEAR.
Y2000
Y2000 IF (LOW-CENTURY)
Y2000 AND (LOW-YEAR)
Y2000 MOVE 01 TO WS-CC-2
Y2000 GO TO 99998-EXIT.
Y2000
Y2000 IF (HIGH-CENTURY)
Y2000 AND (HIGH-YEAR)
Y2000 MOVE 99 TO WS-CC-2
Y2000 GO TO 99998-EXIT.
Y2000
Y2000 IF WS-YY-2 > 40
Y2000 MOVE 19 TO WS-CC-2
Y2000 ELSE
Y2000 MOVE 20 TO WS-CC-2.
Y2000
Y2000 IF VALID-YEAR
Y2000 GO TO 99998-EXIT.
Y2000
Y2000 MOVE ZEROS TO WS-CCYY-2.
Y2000 GO TO 99998-EXIT.
Y2000
```

YY to CCYY

(Continued Next Page)

Y2000

Y2000 99998-IF-YYMM-1.**YYMM to CCYYMM**

```

Y2000
Y2000      MOVE ZEROS TO WS-CC-1.
Y2000
Y2000      MOVE WS-YY-1 TO WS-YEAR.
X2000      MOVE WS-MM-1 TO WS-MONTH.
Y2000
Y2000      IF (LOW-CENTURY)
Y2000      AND (LOW-YEAR)
Y2000      AND (LOW-MONTH)
Y2000      MOVE 01 TO WS-CC-1
Y2000      GO TO 99998-EXIT.
Y2000
Y2000      IF (HIGH-CENTURY)
Y2000      AND (HIGH-YEAR)
Y2000      AND (HIGH-MONTH)
Y2000      MOVE 99 TO WS-CC-1
Y2000      GO TO 99998-EXIT.
Y2000
Y2000      IF WS-YY-1 > 40
Y2000      MOVE 19 TO WS-CC-1
Y2000      ELSE
Y2000      MOVE 20 TO WS-CC-1.
Y2000
Y2000      IF (VALID-YEAR)
Y2000      AND (VALID-MONTH)
Y2000      GO TO 99998-EXIT.
Y2000
Y2000      MOVE ZEROS TO WS-CCYYMM-1.
Y2000      GO TO 99998-EXIT.
Y2000

```

Y2000 99998-IF-YYMM-2.**YYMM to CCYYMM**

```

Y2000
Y2000      MOVE ZEROS TO WS-CC-2.
Y2000
Y2000      MOVE WS-YY-2 TO WS-YEAR.
Y2000      MOVE WS-MM-2 TO WS-MONTH.
Y2000
Y2000      IF (LOW-CENTURY)
Y2000      AND (LOW-YEAR)
Y2000      AND (LOW-MONTH)
Y2000      MOVE 01 TO WS-CC-2
Y2000      GO TO 99998-EXIT.
Y2000
Y2000      IF (HIGH-CENTURY)
Y2000      AND (HIGH-YEAR)
Y2000      AND (HIGH-MONTH)
Y2000      MOVE 99 TO WS-CC-2
Y2000      GO TO 99998-EXIT.
Y2000
Y2000      IF WS-YY-2 > 40
Y2000      MOVE 19 TO WS-CC-2
Y2000      ELSE
Y2000      MOVE 20 TO WS-CC-2.
Y2000
Y2000      IF (VALID-YEAR)
Y2000      AND (VALID-MONTH)
Y2000      GO TO 99998-EXIT.
Y2000
Y2000      MOVE ZEROS TO WS-CCYYMM-2.
Y2000      GO TO 99998-EXIT.
Y2000

```

(Continued Next Page)

Y2000

Y2000 99998-IF-YYMMDD-1.

Y2000

Y2000 MOVE ZEROS TO WS-CC-1.

Y2000

Y2000 MOVE WS-YY-1 TO WS-YEAR.

Y2000

Y2000 MOVE WS-MM-1 TO WS-MONTH.

Y2000

Y2000 MOVE WS-DD-1 TO WS-DAY.

Y2000

Y2000 IF (LOW-CENTURY)

Y2000 AND (LOW-YEAR)

Y2000 AND (LOW-MONTH)

Y2000 AND (LOW-DAY)

Y2000 MOVE 01 TO WS-CC-1

Y2000 GO TO 99998-EXIT.

Y2000

Y2000 IF (HIGH-CENTURY)

Y2000 AND (HIGH-YEAR)

Y2000 AND (HIGH-MONTH)

Y2000 AND (HIGH-DAY)

Y2000 MOVE 99 TO WS-CC-1

Y2000 GO TO 99998-EXIT.

Y2000

Y2000 IF WS-YY-1 > 40**Y2000 MOVE 19 TO WS-CC-1****Y2000 ELSE****Y2000 MOVE 20 TO WS-CC-1.**

Y2000

Y2000 IF (VALID-YEAR)

Y2000 AND (VALID-MONTH)

Y2000 AND (VALID-DAY)

Y2000 GO TO 99998-EXIT.

Y2000

Y2000 MOVE ZEROS TO WS-CCYYMMDD-1.

Y2000

Y2000 GO TO 99998-EXIT.

Y2000

Y2000 99998-IF-YYMMDD-2.

Y2000

Y2000 MOVE ZEROS TO WS-CC-2.

Y2000

Y2000 MOVE WS-YY-2 TO WS-YEAR.

Y2000

Y2000 MOVE WS-MM-2 TO WS-MONTH.

Y2000

Y2000 MOVE WS-DD-2 TO WS-DAY.

Y2000

Y2000 IF (LOW-CENTURY)

Y2000 AND (LOW-YEAR)

Y2000 AND (LOW-MONTH)

Y2000 AND (LOW-DAY)

Y2000 MOVE 01 TO WS-CC-2

Y2000 GO TO 99998-EXIT.

Y2000

Y2000 IF (HIGH-CENTURY)

Y2000 AND (HIGH-YEAR)

Y2000 AND (HIGH-MONTH)

Y2000 AND (HIGH-DAY)

Y2000 MOVE 99 TO WS-CC-2

Y2000 GO TO 99998-EXIT.

Y2000

Y2000 IF WS-YY-2 > 40**Y2000 MOVE 19 TO WS-CC-2****Y2000 ELSE****Y2000 MOVE 20 TO WS-CC-2.**

Y2000

Y2000 IF (VALID-YEAR)

Y2000 AND (VALID-MONTH)

Y2000 AND (VALID-DAY)

Y2000 GO TO 99998-EXIT.

Y2000

Y2000 MOVE ZEROS TO WS-CCYYMMDD-2.

Y2000

Y2000 GO TO 99998-EXIT.

Y2000

YYMMDD to CCYYMMDD

Existence Check

Range Check

Y2K Expansion

Validity Check

Invalid Date (Default)

YYMMDD to CCYYMMDD

(Continued Next Page)

Y2000

Y2000 99998-IF-MMDDYY-1.

MMDDYY to CCYYMMDD

```
Y2000
Y2000 MOVE ZEROS TO WS-CC-1.
Y2000
Y2000 MOVE WS-YY-3 TO WS-YY-1.
Y2000 MOVE WS-MM-3 TO WS-MM-1.
Y2000 MOVE WS-DD-3 TO WS-DD-1.
Y2000
Y2000 MOVE WS-YY-1 TO WS-YEAR.
Y2000 MOVE WS-MM-1 TO WS-MONTH.
Y2000 MOVE WS-DD-1 TO WS-DAY.
Y2000
Y2000 IF (LOW-CENTURY)
Y2000 AND (LOW-YEAR)
Y2000 AND (LOW-MONTH)
Y2000 AND (LOW-DAY)
Y2000 MOVE 01 TO WS-CC-1
Y2000 GO TO 99998-EXIT.
Y2000
Y2000 IF (HIGH-CENTURY)
Y2000 AND (HIGH-YEAR)
Y2000 AND (HIGH-MONTH)
Y2000 AND (HIGH-DAY)
Y2000 MOVE 99 TO WS-CC-1
Y2000 GO TO 99998-EXIT.
Y2000
Y2000 IF WS-YY-1 > 40
Y2000 MOVE 19 TO WS-CC-1
Y2000 ELSE
Y2000 MOVE 20 TO WS-CC-1.
Y2000
Y2000 IF (VALID-YEAR)
Y2000 AND (VALID-MONTH)
Y2000 AND (VALID-DAY)
Y2000 GO TO 99998-EXIT.
Y2000
Y2000 MOVE ZEROS TO WS-CCYYMMDD-1.
Y2000 GO TO 99998-EXIT.
```

(Continued Next Page)

Y2000

Y2000 **99998-IF-MMDDYY-2.**MMDDYY to CCYYMMDD

```
Y2000
Y2000 MOVE ZEROS TO WS-CC-2.
Y2000
Y2000 MOVE WS-YY-3 TO WS-YY-2.
Y2000 MOVE WS-MM-3 TO WS-MM-2.
Y2000 MOVE WS-DD-3 TO WS-DD-2.
Y2000
Y2000 MOVE WS-YY-2 TO WS-YEAR.
Y2000 MOVE WS-MM-2 TO WS-MONTH.
Y2000 MOVE WS-DD-2 TO WS-DAY.
Y2000
Y2000 IF (LOW-CENTURY)
Y2000 AND (LOW-YEAR)
Y2000 AND (LOW-MONTH)
Y2000 AND (LOW-DAY)
Y2000 MOVE 01 TO WS-CC-2
Y2000 GO TO 99998-EXIT.
Y2000
Y2000 IF (HIGH-CENTURY)
Y2000 AND (HIGH-YEAR)
Y2000 AND (HIGH-MONTH)
Y2000 AND (HIGH-DAY)
Y2000 MOVE 99 TO WS-CC-2
Y2000 GO TO 99998-EXIT.
Y2000
Y2000 IF WS-YY-2 > 40
Y2000 MOVE 19 TO WS-CC-2
Y2000 ELSE
Y2000 MOVE 20 TO WS-CC-2.
Y2000
Y2000 IF (VALID-YEAR)
Y2000 AND (VALID-MONTH)
Y2000 AND (VALID-DAY)
Y2000 GO TO 99998-EXIT.
Y2000
Y2000 MOVE ZEROS TO WS-CCYYMMDD-2.
Y2000 GO TO 99998-EXIT.
```

(Continued Next Page)

Y2000

Y2000 99998-IF-YYMMCY-4.YYMMCY to CCYYMMCY

```

Y2000
Y2000      MOVE ZEROS TO WS-CC-4.
Y2000
Y2000      MOVE WS-YY-4 TO WS-YEAR.
Y2000      MOVE WS-MM-4 TO WS-MONTH.
Y2000
Y2000      IF (LOW-CENTURY)
Y2000      AND (LOW-YEAR)
Y2000      AND (LOW-MONTH)
Y2000          MOVE 01 TO WS-CC-4
Y2000          GO TO 99998-EXIT.
Y2000
Y2000      IF (HIGH-CENTURY)
Y2000      AND (HIGH-YEAR)
Y2000      AND (HIGH-MONTH)
Y2000          MOVE 99 TO WS-CC-4
Y2000          GO TO 99998-EXIT.
Y2000
Y2000      IF WS-YY-4 > 40
Y2000          MOVE 19 TO WS-CC-4
Y2000      ELSE
Y2000          MOVE 20 TO WS-CC-4.
Y2000
Y2000      IF (VALID-YEAR)
Y2000      AND (VALID-MONTH)
Y2000          GO TO 99998-EXIT.
Y2000
Y2000      MOVE ZEROS TO WS-CCYYMMCY-4.
Y2000      GO TO 99998-EXIT.
Y2000

```

Y2000 99998-IF-YYMMCY-5.YYMMCY to CCYYMMCY

```

Y2000
Y2000      MOVE ZEROS TO WS-CC-5.
Y2000
Y2000      MOVE WS-YY-5 TO WS-YEAR.
Y2000      MOVE WS-MM-5 TO WS-MONTH.
Y2000
Y2000      IF (LOW-CENTURY)
Y2000      AND (LOW-YEAR)
Y2000      AND (LOW-MONTH)
Y2000          MOVE 01 TO WS-CC-5
Y2000          GO TO 99998-EXIT.
Y2000
Y2000      IF (HIGH-CENTURY)
Y2000      AND (HIGH-YEAR)
Y2000      AND (HIGH-MONTH)
Y2000          MOVE 99 TO WS-CC-5
Y2000          GO TO 99998-EXIT.
Y2000
Y2000      IF WS-YY-5 > 40
Y2000          MOVE 19 TO WS-CC-5
Y2000      ELSE
Y2000          MOVE 20 TO WS-CC-5.
Y2000
Y2000      IF (VALID-YEAR)
Y2000      AND (VALID-MONTH)
Y2000          GO TO 99998-EXIT.
Y2000
Y2000      MOVE ZEROS TO WS-CCYYMMCY-5.
Y2000      GO TO 99998-EXIT.
Y2000

```

(Continued Next Page)

Y2000

Y2000 99998-IF-YYDDD-7.YYMMDDD (Julian) to CCYYMMDDD

```

Y2000
Y2000      MOVE ZEROS TO WS-CC-7.
Y2000
Y2000      MOVE WS-YY-7 TO WS-YEAR.
X2000      MOVE WS-DDD-7 TO WS-DAYS.
Y2000
Y2000      IF (LOW-CENTURY)
Y2000      AND (LOW-YEAR)
Y2000      AND (LOW-DAYS)
Y2000      MOVE 01 TO WS-CC-7
Y2000      GO TO 99998-EXIT.
Y2000
Y2000      IF (HIGH-CENTURY)
Y2000      AND (HIGH-YEAR)
Y2000      AND (HIGH-DAYS)
Y2000      MOVE 99 TO WS-CC-7
Y2000      GO TO 99998-EXIT.
Y2000
Y2000      IF WS-YY-7 > 40
Y2000      MOVE 19 TO WS-CC-7
Y2000      ELSE
Y2000      MOVE 20 TO WS-CC-7.
Y2000
Y2000      IF (VALID-YEAR)
Y2000      AND (VALID-DAYS)
Y2000      GO TO 99998-EXIT.
Y2000
Y2000      MOVE ZEROS TO WS-CCYYDDD-7.
Y2000      GO TO 99998-EXIT.
Y2000

```

Y2000 99998-IF-YYDDD-8.YYMMDDD (Julian) to CCYYMMDDD

```

Y2000
Y2000      MOVE ZEROS TO WS-CC-8.
Y2000
Y2000      MOVE WS-YY-8 TO WS-YEAR.
Y2000      MOVE WS-DDD-8 TO WS-DAYS.
Y2000
Y2000      IF (LOW-CENTURY)
Y2000      AND (LOW-YEAR)
Y2000      AND (LOW-DAYS)
Y2000      MOVE 01 TO WS-CC-8
Y2000      GO TO 99998-EXIT.
Y2000
Y2000      IF (HIGH-CENTURY)
Y2000      AND (HIGH-YEAR)
Y2000      AND (HIGH-DAYS)
Y2000      MOVE 99 TO WS-CC-8
Y2000      GO TO 99998-EXIT.
Y2000
Y2000      IF WS-YY-8 > 40
Y2000      MOVE 19 TO WS-CC-8
Y2000      ELSE
Y2000      MOVE 20 TO WS-CC-8.
Y2000
Y2000      IF (VALID-YEAR)
Y2000      AND (VALID-DAYS)
Y2000      GO TO 99998-EXIT.
Y2000
Y2000      MOVE ZEROS TO WS-CCYYDDD-8.
Y2000      GO TO 99998-EXIT.
Y2000

```

Y2000 99998-EXIT.

```

Y2000      EXIT.
END

```

Standard “Y2K” Processing — Minimum Requirements

The following are the “minimum” changes that must be made to ALL programs to correctly handle Y2K — in addition to any specific code changes needed to handle “**PROCEDURE DIVISION**” COBOL statements (i.e. “IF”, “COMPUTE”, etc.)

These are laid out below, in the order that they would appear in any given COBOL program source. Some, but not all, of these steps apply to production COBOL **COPYPROCS** (source inserts), as noted below.

1. Reference to Copyproc “WS-DATES” in the local WORKING-STORAGE Section:

WS-DATES contains all Y2K-related variables and should be placed at the beginning of working storage.

If not already in place, the following COPY statement should be inserted as shown.

```

...
        WORKING-STORAGE SECTION.

Y2000  COPY WS-DATES.
...

```

2. Reference to Copyprocs “98-WINDOW” and “99-PIVOT” in the PROCEDURE Division:

98-WINDOW contains all necessary paragraphs to process Y2K renovations. This should be placed at the END of the **Procedure Division** as shown:

```

...
Y2000  COPY 98-WINDOW.

Y2000 *COPY 99-PIVOT.
...

```

98-WINDOW checks local date value. expands 2-4 digit years, performs range/existence checks, etc., taking into account various formats for incoming and outgoing years, dates and cycles.

99-PIVOT (not applicable as of 7/1/99), will allow for database “windowing”, as described in “**Standard Y2K Processing — COBOL Copyprocs**”.

Before inserting the above, the source program should be checked to see if these statements have already been inserted during previous Y2K renovation.

3. Identification of Y2K “Tokens” in WORKING-STORAGE:

All working storage variables relating to **dates** and **date-related** values, including cycle, term, (billing) frequency, and date “differences”, are labeled “**Y2000**” in columns 1-5 as shown.

Additionally, the tool kit can be made to place these in a “token file” and called in for later renovation steps.

```
....
Y2000      05  WS-RUN-DATE.
Y2000          10  WS-RUN-YY          PIC 9(02).
Y2000          10  WS-RUN-MM          PIC 9(02).
Y2000          10  WS-RUN-DD          PIC 9(02).
....
Y2000      05  WS-PIA-EFF-YY          PIC S9(05) VALUE ZERO COMP.
Y2000      05  WS-PIA-EFF-MM          PIC S9(05) VALUE ZERO COMP.
Y2000      05  WS-PIA-EFF-DD          PIC S9(05) VALUE ZERO COMP.
....
Y2000          10  WS-BTERM            PIC S9(03)      VALUE ZERO.
....
Y2000      01  WS-DATE-AREAS.
Y2000          05  WS-NEXT-DATE.
Y2000          10  WS-NEXTYY          PIC 99.
Y2000          10  WS-NEXTMM          PIC 99.
Y2000          05  WS-EFFECTIVE-DATE.
Y2000          10  WS-EFFMM          PIC 9(02).
Y2000          10  WS-EFFDD          PIC 9(02).
Y2000          10  WS-EFFYY          PIC 9(02).
Y2000          05  WS-TEST-YYMM.
Y2000          10  WS-TEST-YY          PIC 9(02) COMP VALUE 0.
Y2000          10  WS-TEST-MM          PIC 9(02) COMP VALUE 0.
....
```

Note: The above should be performed on ALL working storage production COPYPROCS.
For example, “WDTEJG”.

4. Labeling of Y2K “Target Statements” in the PROCEDURE DIVISION:

Similarly, all lines of code in the Procedure Division that contain variables relating to **dates** and **date-related** values, including cycle, term, (billing) frequency, and date “differences”, are labeled “**Y2000**” as shown:

```
....
Y2000      ACCEPT WS-RUN-DATE FROM DATE.
....
Y2000      IF  LSM060-LTERM > 35
....
Y2000      IF  ((LSM060-BILFREQ (LSM060-RNTNO) = 'A' OR 'S')
....
Y2000      MOVE LSM060-PBI-INVOICE-DATEYY TO WS-PIA-INV-YY.
Y2000      MOVE LSM060-PBI-INVOICE-DATEMM TO WS-PIA-INV-MM.
Y2000      MOVE LSM060-PBI-INVOICE-DATEDD TO WS-PIA-INV-DD.
....
Y2000      IF  WS-PIA-INV-DD IS NOT EQUAL TO WS-PIA-EFF-DD
....
Y2000      MOVE +31 TO WS-PIA-ANS-DD.
Y2000      SUBTRACT WS-PIA-INV-DD FROM WS-PIA-ANS-DD.
Y2000      ADD WS-PIA-EFF-DD TO WS-PIA-ANS-DD.
....
Y2000      MOVE RTP-REPORT-DATEMM TO WS-TEST-MM.
Y2000      MOVE RTP-REPORT-DATEYY TO WS-TEST-YY.
....
Y2000      IF  WS-TEST-MM > 12
Y2000          SUBTRACT 12 FROM WS-TEST-MM
Y2000          ADD 1 TO WS-TEST-YY.
....
Y2000      IF  LSM060-CYCLE = 2
Y2000          MOVE 30 TO WS-EFFDD
....
```

Note: The above should be performed on ALL production COPYPROCS. For example, "PDTEJG".

Note: The COBOL renovations described above represent the "minimum requirements" for Y2K, and should be performed in all COBOL sources.

FURTHER RECOMMENDATIONS

Once the "minimum" steps outlined above have been performed, the following additional procedures are strongly recommended:

- A. Compile the renovated source copy in a test region and handle any compile errors.
- B. Check for duplicate variable references between the Y2K variables and previously used "local" variables (usually generates a "SEV. M" level error in a compile).

When such conflicts occur, change the local variable names to conform to the following standard in order to eliminate the conflict:

1. **Program** = **WS**-<program name>
2. **Character** = **WSC**-<literal>
3. **String** = **WSL**-<literal string>
4. **Numerics** = **WSN**-<numeric value>.

For examples, change local variable **WS-A** to **WSC-A** in the local program code.

- C. Perform a file compare ("FC") between the latest production copy of the program source (in PBCC*SOURCE), and the test version (the source that is being worked on). Make sure the test version has all recent production changes, including Y2K renovations.
- D. Make any necessary Y2K renovations to the program, using the procedures outlined in this manual.

Once these steps have been completed, the **Y2K Testing Procedures** should then be performed for each COBOL program being updated, if necessary.

“IF” Statement

The following are examples, taken from actual COBOL source programs, of Y2K renovations necessary to correctly process “20xx” dates:

Code Example	Example Identification
Two 6-Digit Dates (YYMMDD)	04 — A
Two 4-Digit Dates (YYMM)	04 — B
Two 2-Digit Years (YY)	04 — C
Two 2-Digit Years “(YY + 1)”	04 — D
Two 2-Digit Years “(YY — 1)”	04 — E
IF/AND Construct (YYMMDD)	04 — F
IF/OR Construct (YYMMDD)	04 — G
IF/AND/OR Construct (YYMMDD)	04 — H
	04 — ____
	04 — ____
	04 — ____
	04 — ____
	04 — ____
	04 — ____

“IF” Statement

Two 6-Digit Dates (YYMMDD)

```

===== SOURCE PROGRAM: LSEE
=====

===== WORKING-STORAGE:
=====
...
...
01 WS-PBCC-MODIFICATIONS-WS.
...
Y2000 05 WS-NEXT-POST-DATE.
Y2000 10 WS-NEXT-POST-DATEYY PIC 99.
Y2000 10 WS-NEXT-POST-DATEMM PIC 99.
Y2000 10 WS-NEXT-POST-DATEDD PIC 99.
...
01 LSE616.
05 LSE616-TRANS-DATA.
10 LSE616-COMMENDATE.
15 LSE616-COMMENDATE-YY PIC 9(02).
15 LSE616-COMMENDATE-MM PIC 9(02).
15 LSE616-COMMENDATE-DD PIC 9(02).
...
...

===== PROCEDURE DIVISION:
=====

===== BEFORE Y2K RENOVATION:
=====
...
...
Y2000 IF LSE616-COMMENDATE > WS-NEXT-POST-DATE
Y2000 MOVE R071 TO ERROR-NO
Y2000 GO TO 30304-COMMENDATE-ERROR.
...
...

===== AFTER Y2K RENOVATION:
=====
...
...
Y2000 * IF LSE616-COMMENDATE > WS-NEXT-POST-DATE
Y2000
Y2000 MOVE LSE616-COMMENDATE TO WS-YYMMDD-1.
Y2000 PERFORM 99998-IF-YYMMDD-1 THRU 99998-EXIT.
Y2000 MOVE WS-NEXT-POST-DATE TO WS-YYMMDD-2.
Y2000 PERFORM 99998-IF-YYMMDD-2 THRU 99998-EXIT.
Y2000
Y2000 IF WS-CCYYMMDD-1 > WS-CCYYMMDD-2
Y2000 MOVE R071 TO ERROR-NO
Y2000 GO TO 30304-COMMENDATE-ERROR.
...
...
=====

```

Business Logic

Y2K Changes

“IF” Statement

Two 4-Digit Dates (YYMM)

```
===== SOURCE PROGRAM: INQ401
=====
```

```
===== WORKING-STORAGE:
=====
```

```
...
...
Y2000 05 WS-WORKDATE.
Y2000 10 WS-WORKDATEYY PIC 9(02).
Y2000 10 WS-WORKDATEMM PIC 9(02).
Y2000 05 WS-SAVE-DATE PIC 9(6).
...
Y2000 05 SD-30DAY-DATE.
Y2000 10 SD-30DAY-DATEMM PIC 9(02).
Y2000 10 FILLER PIC X(01) VALUE '-'.
Y2000 10 SD-30DAY-DATEDD PIC 9(02).
Y2000 10 FILLER PIC X(01) VALUE '-'.
Y2000 10 SD-30DAY-DATEYY PIC 9(02).
...
01 LSQ060.
05 LSQ060--DATA.
Y2000 10 LSQ060-NEXTBILL.
Y2000 15 LSQ060-NEXTBILLYY PIC 9(02).
Y2000 15 LSQ060-NEXTBILLMM PIC 9(02).
Y2000 10 LSQ060-BTERM PIC S9(03)
...
...

```

```
===== PROCEDURE DIVISION:
=====
```

```
===== BEFORE Y2K RENOVATION:
=====
```

```
...
...
Y2000 MOVE SD-30DAY-DATEMM TO WS-WORKDATEMM.
Y2000 MOVE SD-30DAY-DATEYY TO WS-WORKDATEYY.
...
Y2000 IF LSQ060-NEXTBILL IS NOT LESS THAN WS-WORKDATE
GO TO 00400-COMPUTE-UNEARNED.
...
...

```

Business Logic

```
===== AFTER Y2K RENOVATION:
=====
```

```
...
...
Y2000 MOVE SD-30DAY-DATEMM TO WS-WORKDATEMM.
Y2000 MOVE SD-30DAY-DATEYY TO WS-WORKDATEYY.
...
Y2000 * IF LSQ060-NEXTBILL IS NOT LESS THAN WS-WORKDATE
Y2000 MOVE LSQ060-NEXTBILL TO WS-YYMM-1.
Y2000 PERFORM 99998-IF-YYMM-1 THRU 99998-EXIT.
Y2000 MOVE WS-WORKDATE TO WS-YYMM-2.
Y2000 PERFORM 99998-IF-YYMM-2 THRU 99998-EXIT.
Y2000 IF WS-CCYYMM-1 NOT < WS-CCYYMM-2
Y2000 GO TO 00400-COMPUTE-UNEARNED.
...
...

```

Y2K Changes

```
=====
```

“IF” Statement

Two 2-Digit Years (YY)

```
===== SOURCE PROGRAM: LSKD
=====
```

```
===== WORKING-STORAGE:
=====
```

```
...
...
01 LSM060.
1011 03 LSM060-SCHEDULE-INFO.
1011 05 LSM060-LEASE-GENERAL.
...
Y2000 10 LSM060-COMMENDATE-DTE.
Y2000 15 LSM060-COMMENDATE-CEN PIC 9(01).
Y2000 15 LSM060-COMMENDATE.
Y2000 20 LSM060-COMMENDATEYY PIC 9(02).
Y2000 20 LSM060-COMMENDATEMM PIC 9(02).
Y2000 20 LSM060-COMMENDATEDD PIC 9(02).
...
...
01 LSP010.
Y2000 05 RTP-REPORT-DATE.
Y2000 10 RTP-REPORT-DATEYY PIC 9(02).
Y2000 10 RTP-REPORT-DATEMM PIC 9(02).
Y2000 10 RTP-REPORT-DATEDD PIC 9(02).
...
...
===== PROCEDURE DIVISION:
=====
```

```
===== BEFORE Y2K RENOVATION:
=====
```

```
...
...
046800 20400-ACCUM-POSTING-CONT.
Y2000 IF LSM060-COMMENDATEYY > RTP-REPORT-DATEYY
Y2000 GO TO 20401.
...
...
===== AFTER Y2K RENOVATION:
=====
```

Business Logic

```
...
...
046800 20400-ACCUM-POSTING-CONT.
Y2000 * IF LSM060-COMMENDATEYY > RTP-REPORT-DATEYY
Y2000
Y2000 MOVE LSM060-COMMENDATEYY TO WS-YY-1.
Y2000 PERFORM 99998-IF-YY-1 THRU 99998-EXIT.
Y2000 MOVE RTP-REPORT-DATEYY TO WS-YY-2.
Y2000 PERFORM 99998-IF-YY-2 THRU 99998-EXIT.
Y2000
Y2000 IF WS-CCYY-1 > WS-CCYY-2
Y2000 GO TO 20401.
...
...
=====
```

Y2K Changes

“IF” Statement

Two 2-Digit Years “(YY + 1)”

```
===== SOURCE PROGRAM: LSAA
=====
```

```
===== WORKING-STORAGE:
=====
```

```
...
...
011600 01 WORK-AREA.
Y2000 05 DATE-VALIDATE.
Y2000 10 DATE-MM PIC 99.
Y2000 10 DATE-DD PIC 99.
Y2000 10 DATE-YY PIC 99.
...
...
01 LSP010.
Y2000 05 RTP-MF-DATE.
Y2000 10 RTP-MF-DATEYY PIC 9(02).
Y2000 10 RTP-MF-DATEMM PIC 9(02).
Y2000 10 RTP-MF-DATEDD PIC 9(02).
...
...

```

```
===== PROCEDURE DIVISION:
=====
```

```
===== BEFORE Y2K RENOVATION:
=====
```

```
...
...
Y2000 IF (DATE-YY = RTP-MF-DATEYY + 1)
Y2000 AND (DATE-MM = 01)
Y2000 GO TO 10090-VALID-RPT-DTE.
...
...

```

Business Logic

```
===== AFTER Y2K RENOVATION:
=====
```

```
...
...
Y2000 * IF (DATE-YY = RTP-MF-DATEYY + 1)
Y2000 MOVE DATE-YY TO WS-YY-1.
Y2000 PERFORM 99998-IF-YY-1 THRU 99998-EXIT.
Y2000 MOVE RTP-MF-DATEYY TO WS-YY-2.
Y2000 PERFORM 99998-IF-YY-2 THRU 99998-EXIT.
Y2000 IF (WS-CCYY-1 = WS-CCYY-2 + 1)
Y2000 AND (DATE-MM = 01)
Y2000 GO TO 10090-VALID-RPT-DTE.
...
...
=====
```

Y2K Changes

“IF” Statement

Two 2-Digit Years “(YY — 1)”

```
===== SOURCE PROGRAM: LSP230
=====
```

```
===== WORKING-STORAGE:
=====
```

```
...
...
01 LSR230.
  380 05 LSR230-RECORD-DATA.
Y2000 10 LSR230-COMMENDATE.
Y2000 15 LSR230-COMMENDATE-YY PIC 99.
Y2000 15 LSR230-COMMENDATE-MM PIC 99.
Y2000 15 LSR230-COMMENDATE-DD PIC 99.
...
...
01 LSP010.
Y2000 05 RTP-MF-DATE.
Y2000 10 RTP-MF-DATEYY PIC 9(02).
Y2000 10 RTP-MF-DATEMM PIC 9(02).
Y2000 10 RTP-MF-DATEDD PIC 9(02).
...
...
804 01 WORK-ID PIC 9(02) VALUE ZERO.
...
...
=====
```

```
===== PROCEDURE DIVISION:
=====
```

```
===== BEFORE Y2K RENOVATION:
=====
```

```
...
...
Y2000 IF LSR230-COMMENDATE-YY = (RTP-REPORT-DATEYY - 1)
804 SUBTRACT 36 FROM WORK-ID.
...
...
=====
```

Business Logic

```
===== AFTER Y2K RENOVATION:
=====
```

```
...
...
Y2000 * IF LSR230-COMMENDATE-YY = (RTP-REPORT-DATEYY - 1)
Y2000 MOVE LSR230-COMMENDATE-YY TO WS-YY-1.
Y2000 PERFORM 99998-IF-YY-1 THRU 99998-EXIT.
Y2000 MOVE RTP-REPORT-DATEYY TO WS-YY-2.
Y2000 PERFORM 99998-IF-YY-2 THRU 99998-EXIT.
Y2000
Y2000 IF WS-CCYY-1 = (WS-CCYY-2 - 1)
804 SUBTRACT 36 FROM WORK-ID.
...
...
=====
```

Y2K Changes

“IF” Statement

IF/AND Construct (YYMMDD)

```
===== SOURCE PROGRAM: LSLF
=====
```

```
===== WORKING-STORAGE:
=====
```

```
...
Y2000 01 WS-HOLD-DATE.
Y2000 05 WS-HOLD-YY          PIC 99.
Y2000 05 WS-HOLD-MM          PIC 99.
Y2000 05 WS-HOLD-DD          PIC 99.
...
005900 01 LF-MISCELLANEOUS-FIELDS.
980550 05 WS-640              PIC X(03) VALUE '640'.
...
980550 01 WS-TOTAL-LATE      PIC S9(09)V9(02) COMP.
...
01 LSM060.
1011 05 LSM060-PRIOR-BILLINGS-RENT.
1011 10 LSM060-LASTRENTCASH          PIC SH9(08)V9(02).
Y2000 10 LSM060-LASTRENTDATE-DTE.
Y2000 15 LSM060-LASTRENTDATE-CEN    PIC 9(01).
Y2000 15 LSM060-LASTRENTDATE.
Y2000 20 LSM060-LASTRENTDATEYY      PIC 9(02).
Y2000 20 LSM060-LASTRENTDATEMM      PIC 9(02).
Y2000 20 LSM060-LASTRENTDATEDD      PIC 9(02).
1011 10 LSM060-LASTPAYTYPE.
1011 15 LSM060-LASTPAY-PREFIX        PIC X(02).
1011 15 LSM060-LASTPAY-MEMO         PIC X(01).
...
```

```
===== PROCEDURE DIVISION:
=====
```

```
===== BEFORE Y2K RENOVATION:
=====
```

```
...
Y2000 IF (LSM060-LASTPAYTYPE = WS-640)
Y2000 AND (LSM060-LASTRENTDATE > WS-HOLD-DATE)
Y2000 ADD LSM060-LASTRENTCASH TO WS-TOTAL-LATE.
...
```

Business Logic

```
===== AFTER Y2K RENOVATION:
=====
```

```
...
Y2000
Y2000 MOVE 0 TO WS-Q.
Y2000
Y2000 IF (LSM060-LASTPAYTYPE = WS-640)
Y2000 * AND (LSM060-LASTRENTDATE > WS-HOLD-DATE)
Y2000
Y2000 MOVE 1 TO WS-Q.
Y2000
Y2000 MOVE LSM060-LASTRENTDATE TO WS-YYMMDD-1.
Y2000 PERFORM 99998-IF-YYMMDD-1 THRU 99998-EXIT.
Y2000 MOVE WS-HOLD-DATE TO WS-YYMMDD-2.
Y2000 PERFORM 99998-IF-YYMMDD-2 THRU 99998-EXIT.
Y2000
Y2000 IF (WS-Q = 1)
Y2000 AND (WS-CCYYMMDD-1 > WS-CCYYMMDD-2)
Y2000 ADD LSM060-LASTRENTCASH TO WS-TOTAL-LATE.
...
=====
```

Y2K Changes

“IF” Statement

IF/OR Construct (YYMMDD)

```

===== SOURCE PROGRAM: LSAAA
=====

===== WORKING-STORAGE:
=====
...
...
01 LSX005.
Y2000 05 LSX005-REPORT-DATE.
Y2000 10 LSX005-REPORT-DATEYY PIC 99.
Y2000 10 LSX005-REPORT-DATEMM PIC 99.
Y2000 10 LSX005-REPORT-DATEDD PIC 99.
...
01 LSP010.
Y2000 05 RTP-CF-DATE.
Y2000 10 RTP-CF-DATEYY PIC 9(02).
Y2000 10 RTP-CF-DATEMM PIC 9(02).
Y2000 10 RTP-CF-DATEDD PIC 9(02).
...

===== PROCEDURE DIVISION:
=====

===== BEFORE Y2K RENOVATION:
=====
...
Y2000 IF (LSX005-REPORT-DATE < RTP-MF-DATE)
Y2000 OR (LSX005-REPORT-DATE < RTP-CF-DATE)
Y2000 GO TO 10080-DTE-ERR.
...

===== AFTER Y2K RENOVATION:
=====
...
Y2000 * IF (LSX005-REPORT-DATE < RTP-MF-DATE)
Y2000 MOVE LSX005-REPORT-DATE TO WS-YYMMDD-1.
Y2000 PERFORM 99998-IF-YYMMDD-1 THRU 99998-EXIT.
Y2000 MOVE RTP-MF-DATE TO WS-YYMMDD-2.
Y2000 PERFORM 99998-IF-YYMMDD-2 THRU 99998-EXIT.
Y2000 MOVE 0 TO WS-A.
Y2000 IF (WS-CCYYMMDD-1 < WS-CCYYMMDD-2)
Y2000 * OR (LSX005-REPORT-DATE < RTP-CF-DATE)
Y2000 MOVE 1 TO WS-A.
Y2000 MOVE LSX005-REPORT-DATE TO WS-YYMMDD-1.
Y2000 PERFORM 99998-IF-YYMMDD-1 THRU 99998-EXIT.
Y2000 MOVE RTP-CF-DATE TO WS-YYMMDD-2.
Y2000 PERFORM 99998-IF-YYMMDD-2 THRU 99998-EXIT.
Y2000 IF (WS-A = 1)
Y2000 OR (WS-CCYYMMDD-1 < WS-CCYYMMDD-2)
Y2000 GO TO 10080-DTE-ERR.
...
=====

```

Business Logic

Y2K Changes

“IF” Statement

IF/AND/OR Construct (YYMMDD)

```

===== SOURCE PROGRAM: PTS340
=====

===== WORKING-STORAGE:
=====
...
...
Y2000 01 WS-PRIOR-DATE.
Y2000 05 WS-PR-YY          PIC 99.
Y2000 05 WS-PR-MM          PIC 99.
Y2000 05 WS-PR-DD          PIC 99.
...
...
01 LSM060.
...
1011 05 LSM060-LEASE-GENERAL.
1011 10 LSM060-STATUS          PIC 9(01).
...
1011 05 LSM060-TERMINATION-DATA.
...
Y2000 10 LSM060-TERMDATE-DTE.
Y2000 15 LSM060-TERMDATE-CEN    PIC 9(01).
Y2000 15 LSM060-TERMDATE.
Y2000 20 LSM060-TERMDATEYY    PIC 9(02).
Y2000 20 LSM060-TERMDATEMM    PIC 9(02).
Y2000 20 LSM060-TERMDATEDD    PIC 9(02).
...
05 LSM060-AUDIT-TRAIL.
1011 10 LSM060-KEYCHANGE          PIC 9(01).
Y2000 10 LSM060-MAINTDTE-DTE.
Y2000 15 LSM060-MAINTDTE-CEN    PIC 9(01).
Y2000 15 LSM060-MAINTDTE.
Y2000 20 LSM060-MAINTDTEYY    PIC 9(02).
Y2000 20 LSM060-MAINTDTEMM    PIC 9(02).
Y2000 20 LSM060-MAINTDTEDD    PIC 9(02).
...
...

===== PROCEDURE DIVISION:
=====

===== BEFORE Y2K RENOVATION:
=====
...
...
Y2000 IF (LSM060-STATUS = 5 OR 6)
Y2000 AND ((LSM060-TERMDATE < WS-PRIOR-DATE)
Y2000 OR (LSM060-MAINTDTE < WS-PRIOR-DATE))
Y2000 GO TO 2999-EXIT.
...
...
=====

```

Business Logic

(Continued Next Page)

===== AFTER Y2K RENOVATION:
=====

...
...

Y2K Changes

```
Y2000
Y2000      MOVE 0 TO WS-A.
Y2000
Y2000      IF (LSM060-STATUS = 5 OR 6)
Y2000 *      AND ((LSM060-TERMDATE < WS-PRIOR-DATE)
Y2000 *      OR (LSM060-MAINTDTE < WS-PRIOR-DATE))
Y2000
Y2000      MOVE 1 TO WS-A.
Y2000
Y2000      MOVE LSM060-TERMDATE TO WS-YYMMDD-1.
Y2000      PERFORM 99998-IF-YYMMDD-1 THRU 99998-EXIT.
Y2000      MOVE WS-PRIOR-DATE TO WS-YYMMDD-2.
Y2000      PERFORM 99998-IF-YYMMDD-2 THRU 99998-EXIT.
Y2000
Y2000      MOVE 0 TO WS-B.
Y2000
Y2000      IF (WS-CCYYMMDD-1 < WS-CCYYMMDD-2)
Y2000
Y2000      MOVE 1 TO WS-B.
Y2000
Y2000      MOVE LSM060-MAINTDTE TO WS-YYMMDD-1.
Y2000      PERFORM 99998-IF-YYMMDD-1 THRU 99998-EXIT.
Y2000      MOVE WS-PRIOR-DATE TO WS-YYMMDD-2.
Y2000      PERFORM 99998-IF-YYMMDD-2 THRU 99998-EXIT.
Y2000
Y2000      MOVE 0 TO WS-C.
Y2000
Y2000      IF (WS-CCYYMMDD-1 < WS-CCYYMMDD-2)
Y2000
Y2000      MOVE 1 TO WS-C.
Y2000
Y2000      IF (WS-A = 1)
Y2000      AND ((WS-B = 1)
Y2000      OR (WS-C = 1))
Y2000 *      GO TO 2999-EXIT.
```

...
...
=====

“SUBTRACT” Statement

The following are examples, taken from actual COBOL source programs, of Y2K renovations necessary to correctly process “20xx” dates:

Code Example	Example Identification
Subtract Constant from 2-Digit Year	05 — A
Difference between Two 2-Digit Years (in Years)	05 — B
Difference between Two COMP Values (in Years)	05 — C
	05 — ____
	05 — ____
	05 — ____
	05 — ____
	05 — ____
	05 — ____
	05 — ____
	05 — ____
	05 — ____
	05 — ____
	05 — ____

“SUBTRACT” Statement

Subtract Constant from 2-Digit Year

```

===== SOURCE PROGRAM:  LSXAC
=====
===== WORKING-STORAGE:
=====
...
Y2000 01 WS-SAVE-NEXTYY          PIC 9(02)          VALUE ZERO.
...
...

===== PROCEDURE DIVISION:
=====

===== BEFORE Y2K RENOVATION:
=====
...
...
Y2000      IF LSM060-NEXTBILLMM < RTP-REPORT-DATEMM
Y2000          ADD 12 TO WS-SAVE-NEXTMM
Y2000          SUBTRACT 1 FROM WS-SAVE-NEXTYY.
...
...

===== AFTER Y2K RENOVATION:
=====
...
...
Y2000      IF LSM060-NEXTBILLMM < RTP-REPORT-DATEMM
Y2000          ADD 12 TO WS-SAVE-NEXTMM
Y2000
Y2000 *      SUBTRACT 1 FROM WS-SAVE-NEXTYY.
Y2000
Y2000          MOVE WS-SAVE-NEXTYY TO WS-YY-1
Y2000          PERFORM 99998-IF-YY-1 THRU 99998-EXIT
Y2000          SUBTRACT 1 FROM WS-CCYY-1
Y2000
Y2000          MOVE WS-YY-1 TO WS-SAVE-NEXTYY.
...
...
=====

```

Business Logic

Y2K Changes

“SUBTRACT” Statement

Difference between Two 2-Digit Years (in Years)

```
===== SOURCE PROGRAM: TRU101
=====
```

```
===== WORKING-STORAGE:
=====
```

```
...
...
Y2000 01 WS-COMM-DATE.
Y2000 05 WS-COMM-DATEYY          PIC 9(02).
Y2000 05 WS-COMM-DATEMM          PIC 9(02).
Y2000 01 WS-COMM-DATE-N REDEFINES
Y2000  WS-COMM-DATE              PIC 9(04).
...
... 01 WS-ACCUMS.
Y2000 05 WS-ADD-YEARS             PIC S9(03).
Y2000 05 WS-ADD-MONTHS           PIC S9(03).
Y2000 05 WS-RES-MONTHS           PIC S9(03).
Y2000 05 WS-ADJ-ITERM            PIC S9(03).
Y2000 05 WS-RES-YEARS            PIC S9(03).
Y2000 05 WS-MAINT-YEARS          PIC S9(03).
Y2000 05 WS-COMM-YEARS           PIC S9(03).
Y2000 05 WS-COMM-MONTHS          PIC S9(03).
...
... 01 LSP010.
Y2000 05 RTP-PRIOR-MF-REPORT-DATE.
Y2000 10 RTP-PRIOR-MF-REPORT-DATEYY PIC 9(02).
Y2000 10 RTP-PRIOR-MF-REPORT-DATEMM PIC 9(02).
Y2000 10 RTP-PRIOR-MF-REPORT-DATEDD PIC 9(02).
...
...

```

```
===== PROCEDURE DIVISION:
=====
```

```
===== BEFORE Y2K RENOVATION:
=====
```

```
...
...
Y2000 11030-CALC-MONTHS.
Y2000  MOVE ZEROES TO WS-ADD-MONTHS.
Y2000  SUBTRACT RTP-PRIOR-MF-REPORT-DATEYY FROM WS-COMM-DATEYY
Y2000  GIVING WS-COMM-YEARS.
...
...
=====
```

Business Logic

(Continued Next Page)

```

===== AFTER Y2K RENOVATION:
=====
...
...
Y2000 11030-CALC-MONTHS.
Y2000 MOVE ZEROES TO WS-ADD-MONTHS.

Y2000
Y2000 * SUBTRACT RTP-PRIOR-MF-REPORT-DATEYY FROM WS-COMM-DATEYY
Y2000 * GIVING WS-COMM-YEARS.
Y2000
Y2000 MOVE RTP-PRIOR-MF-REPORT-DATEYY TO WS-YY-1.
Y2000 PERFORM 99998-IF-YY-1 THRU 99998-EXIT.
Y2000
Y2000 IF WS-CCYY-1 > 1900
Y2000 SUBTRACT 1900 FROM WS-CCYY-1.
Y2000
Y2000 MOVE WS-COMM-DATEYY TO WS-YY-2.
Y2000 PERFORM 99998-IF-YY-2 THRU 99998-EXIT.
Y2000
Y2000 IF WS-CCYY-2 > 1900
Y2000 SUBTRACT 1900 FROM WS-CCYY-2.
Y2000
Y2000 SUBTRACT WS-CCYY-1 FROM WS-CCYY-2
Y2000 GIVING WS-COMM-YEARS.
...
...
=====

```

Y2K Changes

“SUBTRACT” Statement

Difference between Two COMP Values (in Years)

```

===== SOURCE PROGRAM: INQ202
=====

===== WORKING-STORAGE:
=====
...
Y2000 01 WS-DATES-LIST.
Y2000 05 WS-ED-CURNT-DATE.
Y2000 10 WS-ED-CURNT-MM      PIC 9(02).
2527 10 FILLER              PIC X(01) VALUE '-'.
Y2000 10 WS-ED-CURNT-DD      PIC 9(02).
2527 10 FILLER              PIC X(01) VALUE '-'.
Y2000 10 WS-ED-CURNT-YY      PIC 9(02).
...
Y2000 01 WS-PCNT-INV-AMT-WORK-AREA.
Y2000 05 WS-PIA-EFF-YY      PIC S9(05) VALUE ZERO COMP.
Y2000 05 WS-PIA-EFF-MM      PIC S9(05) VALUE ZERO COMP.
Y2000 05 WS-PIA-EFF-DD      PIC S9(05) VALUE ZERO COMP.
Y2000 05 WS-PIA-INV-YY      PIC S9(05) VALUE ZERO COMP.
Y2000 05 WS-PIA-INV-MM      PIC S9(05) VALUE ZERO COMP.
Y2000 05 WS-PIA-INV-DD      PIC S9(05) VALUE ZERO COMP.
Y2000 05 WS-PIA-ANS-YY      PIC S9(05) VALUE ZERO COMP.
Y2000 05 WS-PIA-ANS-MM      PIC S9(05) VALUE ZERO COMP.
Y2000 05 WS-PIA-ANS-DD      PIC S9(05) VALUE ZERO COMP.
Y2000 05 WS-PIA-ANS        PIC S9(07) VALUE ZERO COMP.
...
Y2000 01 LSQ060.
Y2000 05 LSQ060-DATA.
...
Y2000 10 LSQ060-INVOICE-DATE.
Y2000 15 LSQ060-INVOICE-DATEYY      PIC 9(02).
Y2000 15 LSQ060-INVOICE-DATEMM      PIC 9(02).
Y2000 15 LSQ060-INVOICE-DATEDDD     PIC 9(02).
...
===== PROCEDURE DIVISION:
=====

===== BEFORE Y2K RENOVATION:
=====
...
Y2000 MOVE LSQ060-INVOICE-DATEYY TO WS-PIA-INV-YY.
Y2000 MOVE LSQ060-INVOICE-DATEMM TO WS-PIA-INV-MM.
Y2000 MOVE LSQ060-INVOICE-DATEDDD TO WS-PIA-INV-DD.
Y2000 MOVE WS-ED-CURNT-YY TO WS-PIA-EFF-YY.
Y2000 MOVE WS-ED-CURNT-MM TO WS-PIA-EFF-MM.
Y2000 MOVE WS-ED-CURNT-DD TO WS-PIA-EFF-DD.
...
2527 01235-CALC-TRADE-IN.
Y2000 SUBTRACT WS-PIA-INV-YY FROM WS-PIA-EFF-YY
Y2000 GIVING WS-PIA-ANS-YY.
Y2000 SUBTRACT WS-PIA-INV-MM FROM WS-PIA-EFF-MM
Y2000 GIVING WS-PIA-ANS-MM.
...
=====

```

Business Logic

(Continued Next Page)

```

===== AFTER Y2K RENOVATION:
=====
...
...
Y2000 MOVE LSQ060-INVOICE-DATEYY TO WS-PIA-INV-YY.
Y2000 MOVE LSQ060-INVOICE-DATEMM TO WS-PIA-INV-MM.
Y2000 MOVE LSQ060-INVOICE-DATEDD TO WS-PIA-INV-DD.
Y2000 MOVE WS-ED-CURNT-YY TO WS-PIA-EFF-YY.
Y2000 MOVE WS-ED-CURNT-MM TO WS-PIA-EFF-MM.
Y2000 MOVE WS-ED-CURNT-DD TO WS-PIA-EFF-DD.
...
...
2527 01235-CALC-TRADE-IN.
Y2000
Y2000 * SUBTRACT WS-PIA-INV-YY FROM WS-PIA-EFF-YY
Y2000 * GIVING WS-PIA-ANS-YY.
Y2000
Y2000 MOVE WS-PIA-EFF-YY TO WS-YY-6-COMP.
Y2000 MOVE WS-YY-6 TO WS-YY-1.
Y2000 PERFORM 99998-IF-YY-1 THRU 99998-EXIT.
Y2000
Y2000 MOVE WS-PIA-INV-YY TO WS-YY-6-COMP.
Y2000 MOVE WS-YY-6 TO WS-YY-2.
Y2000 PERFORM 99998-IF-YY-2 THRU 99998-EXIT.
Y2000
Y2000 SUBTRACT WS-CCYY-2 FROM WS-CCYY-1
Y2000 GIVING WS-PIA-ANS-YY.
Y2000
Y2000 SUBTRACT WS-PIA-INV-MM FROM WS-PIA-EFF-MM
Y2000 GIVING WS-PIA-ANS-MM.
...
...
=====

```

Y2K Changes